Rules for integrands of the form $(f x)^m (d + e x^2)^p (a + b \text{ArcCosh}[c x])^n$

1. $\int (f x)^m (d + e x^2)^p (a + b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d + e == 0$

0: $\int (f x)^m (d1 + e1 x)^p (d2 + e2 x)^p (a + b \text{ArcCosh}[c x])^n \, dx$ when $d2\, e1 + d1\, e2 == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $d2\, e1 + d1\, e2 == 0$, then $(d1 + e1 x)(d2 + e2 x) == d1\, d2 + e1\, e2\, x^2$

Rule: If $d2\, e1 + d1\, e2 == 0 \land p \in \mathbb{Z}$, then

$$\int (f x)^m (d1 + e1 x)^p (d2 + e2 x)^p (a + b \text{ArcCosh}[c x])^n \, dx \longrightarrow \int (f x)^m (d1\, d2 + e1\, e2\, x^2)^p (a + b \text{ArcCosh}[c x])^n \, dx$$

–

Program code:

```
Int[(f_.*x_)^m_.*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[(f*x)^m*(d1*d2+e1*e2*x^2)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,n},x] && EqQ[d2*e1+d1*e2,0] && IntegerQ[p]
```

1. $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $c^2 d+e=0\land n>0$

   1. $\int x \left(d+e\,x^2\right)^p (a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $c^2 d+e=0\land n>0$

      1: $\int \frac{x\,(a+b\,\text{ArcCosh}[c\,x])^n}{d+e\,x^2}\,dx$ when $c^2 d+e=0\land n\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $c^2 d+e=0$, then $\frac{x}{d+e\,x^2}=\frac{1}{e}\,\text{Subst}[\text{Coth}[x],\,x,\,\text{ArcCosh}[c\,x]]\,\partial_x\text{ArcCosh}[c\,x]$

Basis: If $c^2 d+e=0$, then $\frac{x}{d+e\,x^2}=-\frac{1}{b\,e}\,\text{Subst}\left[\text{Coth}\left[\frac{a}{b}-\frac{x}{b}\right],\,x,\,a+b\,\text{ArcCosh}[c\,x]\right]\,\partial_x\left(a+b\,\text{ArcCosh}[c\,x]\right)$

Note: If $n\in\mathbb{Z}^+$, then $(a+b\,x)^n\,\text{Coth}[x]$ is integrable in closed-form.

Rule: If $c^2 d+e=0\land n\in\mathbb{Z}^+$, then

$$\int \frac{x\,(a+b\,\text{ArcCosh}[c\,x])^n}{d+e\,x^2}\,dx\;\longrightarrow\;\frac{1}{e}\,\text{Subst}\left[\int (a+b\,x)^n\,\text{Coth}[x]\,dx,\,x,\,\text{ArcCosh}[c\,x]\right]$$

Program code:

```
Int[x_*(a_.+b_.*ArcCosh[c_.*x_])^n_./(d_+e_.*x_^2),x_Symbol] :=
  1/e*Subst[Int[(a+b*x)^n*Coth[x],x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[n,0]
```

2: $\int x\left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \,dx$ when $c^2 d+e=0 \land n>0 \land p\neq -1$

Derivation: Integration by parts and piecewise constant extraction

Basis: $x\left(d+e x^2\right)^p = \partial_x \frac{\left(d+e x^2\right)^{p+1}}{2 e (p+1)}$

Basis: If $e1=c\,d1 \land e2=-c\,d2 \land p\neq -1$, then $x\,(d1+e1\,x)^p\,(d2+e2\,x)^p = \partial_x \frac{(d1+e1\,x)^{p+1}\,(d2+e2\,x)^{p+1}}{2\,e1\,e2\,(p+1)}$

Basis: $\partial_x (a+b \operatorname{ArcCosh}[c x])^n = \frac{b c n (a+b \operatorname{ArcCosh}[c x])^{n-1}}{\sqrt{1+c x}\,\sqrt{-1+c x}}$

Basis: If $c^2 d+e=0$, then $\partial_x \frac{\left(d+e x^2\right)^p}{(1+c x)^p\,(-1+c x)^p} = 0$

Rule: If $c^2 d+e=0 \land n>0 \land p\neq -1$, then

$$\int x\left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \,dx$$

$$\longrightarrow \frac{\left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcCosh}[c x])^n}{2 e (p+1)}-\frac{b c n}{2 e (p+1)} \int \frac{\left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcCosh}[c x])^{n-1}}{\sqrt{1+c x}\,\sqrt{-1+c x}} \,dx$$

$$\longrightarrow \frac{\left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcCosh}[c x])^n}{2 e (p+1)}-\frac{b n \left(d+e x^2\right)^p}{2 c (p+1) (1+c x)^p (-1+c x)^p} \int (1+c x)^{p+\frac{1}{2}} (-1+c x)^{p+\frac{1}{2}} (a+b \operatorname{ArcCosh}[c x])^{n-1} \,dx$$

\-

Program code:

```
Int[x_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e*(p+1)) -
  b*n/(2*c*(p+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && NeQ[p,-1]
```

```
Int[x_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e1*e2*(p+1)) -
  b*n/(2*c*(p+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,p},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && NeQ[p,-1]
```

2. $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcCosh}[c\,x])^n \,dx$ when $c^2\,d+e==0\,\land\,n>0\,\land\,m+2\,p+3==0$

1: $\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{x\left(d+e\,x^2\right)}\,dx$ when $c^2\,d+e==0\,\land\,n\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $c^2\,d+e==0$, then $\frac{1}{x\left(d+e\,x^2\right)}==-\frac{1}{d}\,\text{Subst}\left[\frac{1}{\text{Cosh}[x]\,\text{Sinh}[x]},\,x,\,\text{ArcCosh}[c\,x]\right]\partial_x\text{ArcCosh}[c\,x]$

Basis: If $c^2\,d+e==0$, then $\frac{1}{x\left(d+e\,x^2\right)}==-\frac{1}{b\,d}\,\text{Subst}\left[\frac{1}{\text{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]\,\text{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right]},\,x,\,a+b\,\text{ArcCosh}[c\,x]\right]\partial_x\left(a+b\,\text{ArcCosh}[c\,x]\right)$

Rule: If $c^2\,d+e==0\,\land\,n\in\mathbb{Z}^+$, then

$$\int \frac{(a+b\,\text{ArcCosh}[c\,x])^n}{x\left(d+e\,x^2\right)}\,dx\;\longrightarrow\;-\frac{1}{d}\,\text{Subst}\left[\int \frac{(a+b\,x)^n}{\text{Cosh}[x]\,\text{Sinh}[x]}\,dx,\,x,\,\text{ArcCosh}[c\,x]\right]$$

Program code:

```
Int[(a_.+b_.*ArcCosh[c_.*x_])^n_./(x_*(d_+e_.*x_^2)),x_Symbol] :=
  -1/d*Subst[Int[(a+b*x)^n/(Cosh[x]*Sinh[x]),x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[n,0]
```

2: $\int (f\,x)^m \left(d+e\,x^2\right)^p \,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $c^2\,d+e=0\,\land\,n>0\,\land\,m+2\,p+3=0\,\land\,m\neq-1$

Derivation: Integration by parts and piecewise constant extraction

Basis: If $m+2\,p+3=0$, then $(f\,x)^m \left(d+e\,x^2\right)^p = \partial_x \frac{(f\,x)^{m+1}\left(d+e\,x^2\right)^{p+1}}{d\,f\,(m+1)}$

Basis: If $d2\,e1+d1\,e2=0\,\land\,m+2\,p+3=0$, then $(f\,x)^m\,(d1+e1\,x)^p\,(d2+e2\,x)^p = \partial_x \frac{(f\,x)^{m+1}\,(d1+e1\,x)^{p+1}\,(d2+e2\,x)^{p+1}}{d1\,d2\,f\,(m+1)}$

Basis: $\partial_x\,(a+b\,\text{ArcCosh}[c\,x])^n = \frac{b\,c\,n\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}$

Basis: If $c^2\,d+e=0$, then $\partial_x \frac{\left(d+e\,x^2\right)^p}{(1+c\,x)^p\,(-1+c\,x)^p} = 0$

Rule: If $c^2\,d+e=0\,\land\,n>0\,\land\,m+2\,p+3=0\,\land\,m\neq-1$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcCosh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{(f\,x)^{m+1}\left(d+e\,x^2\right)^{p+1}(a+b\,\text{ArcCosh}[c\,x])^n}{d\,f\,(m+1)}+$$

$$\frac{b\,c\,n\left(d+e\,x^2\right)^p}{f\,(m+1)\,(1+c\,x)^p\,(-1+c\,x)^p}\int (f\,x)^{m+1}\,(1+c\,x)^{p+\frac{1}{2}}\,(-1+c\,x)^{p+\frac{1}{2}}\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(d*f*(m+1)) +
  b*c*n/(f*(m+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && EqQ[m+2*p+3,0] && NeQ[m,-1]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(d1*d2*f*(m+1)) +
  b*c*n/(f*(m+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,p},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && EqQ[m+2*p+3,0] && NeQ[p,-1]
```

3. $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \,dx$ when $c^2 d+e=0 \land n>0 \land p>0$

1. $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x]) \,dx$ when $c^2 d+e=0 \land p>0$

1. $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x]) \,dx$ when $c^2 d+e=0 \land p\in\mathbb{Z}^+$

1. $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x]) \,dx$ when $c^2 d+e=0 \land p\in\mathbb{Z}^+ \land \frac{m-1}{2}\in\mathbb{Z}^-$

1: $\int \frac{\left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])}{x} \,dx$ when $c^2 d+e=0 \land p\in\mathbb{Z}^+$

Derivation: Inverted integration by parts

Rule: If $c^2 d+e=0 \land p\in\mathbb{Z}^+$, then

$$\int \frac{\left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])}{x} \,dx \longrightarrow$$
$$\frac{\left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])}{2p} - \frac{b c (-d)^p}{2p} \int (1+c x)^{p-\frac{1}{2}} (-1+c x)^{p-\frac{1}{2}} \,dx + d \int \frac{\left(d+e x^2\right)^{p-1} (a+b \operatorname{ArcCosh}[c x])}{x} \,dx$$

—

Program code:

```
Int[(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])/x_,x_Symbol] :=
  (d+e*x^2)^p*(a+b*ArcCosh[c*x])/(2*p) -
  b*c*(-d)^p/(2*p)*Int[(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2),x] +
  d*Int[(d+e*x^2)^(p-1)*(a+b*ArcCosh[c*x])/x,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[p,0]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x]) \,dx$ when $c^2 d+e=0 \land p\in\mathbb{Z}^+ \land \frac{m+1}{2}\in\mathbb{Z}^-$

Derivation: Inverted integration by parts

Rule: If $c^2 d+e=0 \land p\in\mathbb{Z}^+ \land \frac{m+1}{2}\in\mathbb{Z}^-$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x]) \, dx \longrightarrow \frac{(f x)^{m+1} \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])}{f (m+1)} - \frac{b c (-d)^p}{f (m+1)} \int (f x)^{m+1} (1+c x)^{p-\frac{1}{2}} (-1+c x)^{p-\frac{1}{2}} dx - \frac{2 e p}{f^2 (m+1)} \int (f x)^{m+2} \left(d+e x^2\right)^{p-1} (a+b \operatorname{ArcCosh}[c x]) \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcCosh[c*x])/(f*(m+1)) -
  b*c*(-d)^p/(f*(m+1))*Int[(f*x)^(m+1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2),x] -
  2*e*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^(p-1)*(a+b*ArcCosh[c*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[c^2*d+e,0] && IGtQ[p,0] && ILtQ[(m+1)/2,0]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x]) \, dx$ when $c^2 d+e=0 \land p \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $c^2 d+e=0 \land p \in \mathbb{Z}^+$, let $u \to \int (f x)^m \left(d+e x^2\right)^p dx$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x]) \, dx \longrightarrow u \, (a+b \operatorname{ArcCosh}[c x]) - b c \int \frac{u}{\sqrt{1+c x} \sqrt{-1+c x}} \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[c^2*d+e,0] && IGtQ[p,0]
```

2: $\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])\, dx$ when $c^2 d+e=0 \wedge p-\frac{1}{2}\in\mathbb{Z} \wedge p\neq-\frac{1}{2} \wedge \left(\frac{m+1}{2}\in\mathbb{Z}^+ \vee \frac{m+2p+3}{2}\in\mathbb{Z}^-\right)$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\partial_x (a+b \operatorname{ArcCosh}[c x]) = \frac{b c}{\sqrt{1+c x}\,\sqrt{-1+c x}}$

Basis: If $c^2 d+e=0$, then $\partial_x \frac{\sqrt{d+e x^2}}{\sqrt{1+c x}\,\sqrt{-1+c x}} = 0$

Note: If $p-\frac{1}{2}\in\mathbb{Z} \wedge \left(\frac{m+1}{2}\in\mathbb{Z}^+ \vee \frac{m+2p+3}{2}\in\mathbb{Z}^-\right)$, then $\int x^m (d+e x^2)^p\, dx$ is an algebraic function not involving logarithms, inverse trig or inverse hyperbolic functions.

Rule: If $c^2 d+e=0 \wedge p-\frac{1}{2}\in\mathbb{Z} \wedge p\neq-\frac{1}{2} \wedge \left(\frac{m+1}{2}\in\mathbb{Z}^+ \vee \frac{m+2p+3}{2}\in\mathbb{Z}^-\right)$, let $u\to\int x^m \left(d+e x^2\right)^p dx$, then

$$\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])\, dx$$

$$\longrightarrow u\,(a+b \operatorname{ArcCosh}[c x]) - b c\int \frac{u}{\sqrt{1+c x}\,\sqrt{-1+c x}}\, dx$$

$$\longrightarrow u\,(a+b \operatorname{ArcCosh}[c x]) - \frac{b c\sqrt{d+e x^2}}{\sqrt{1+c x}\,\sqrt{-1+c x}}\int \frac{u}{\sqrt{d+e x^2}}\, dx$$

Program code:

```
Int[x_^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[x^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcCosh[c*x],u] -
  b*c*Simp[Sqrt[d+e*x^2]/(Sqrt[1+c*x]*Sqrt[-1+c*x])]*Int[SimplifyIntegrand[u/Sqrt[d+e*x^2],x],x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IntegerQ[p-1/2] && NeQ[p,-1/2] && (IGtQ[(m+1)/2,0] || ILtQ[(m+2*p+3)/2,0])
```

```
Int[x_^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[x^m*(d1+e1*x)^p*(d2+e2*x)^p,x]},
  Dist[a+b*ArcCosh[c*x],u] -
  b*c*Simp[Sqrt[d1+e1*x]*Sqrt[d2+e2*x]/(Sqrt[1+c*x]*Sqrt[-1+c*x])]*Int[SimplifyIntegrand[u/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x]),x],x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && IntegerQ[p-1/2] && NeQ[p,-1/2] && (IGtQ[(m+1)/2,0] || ILtQ[(m+2*p+3)/2,0]
```

2. $\int (f\,x)^m \sqrt{d+e\,x^2}\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $c^2 d+e=0\land n>1$

1: $\int (f\,x)^m \sqrt{d+e\,x^2}\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $c^2 d+e=0\land n>0\land m<-1$

Derivation: Inverted integration by parts

Note: The piecewise constant factor in the second integral reduces the degree of d in the resulting antiderivative.

Rule: If $c^2 d+e=0\land n>0\land m<-1$, then

$$\int (f\,x)^m \sqrt{d+e\,x^2}\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{(f\,x)^{m+1}\sqrt{d+e\,x^2}\,(a+b\,\text{ArcCosh}[c\,x])^n}{f\,(m+1)}-$$

$$\frac{b\,c\,n\,\sqrt{d+e\,x^2}}{f\,(m+1)\,\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}\int (f\,x)^{m+1}\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}\,dx-\frac{c^2\sqrt{d+e\,x^2}}{f^2\,(m+1)\,\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}\int \frac{(f\,x)^{m+2}\,(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d+e*x^2]*(a+b*ArcCosh[c*x])^n/(f*(m+1)) -
  b*c*n/(f*(m+1))*Simp[Sqrt[d+e*x^2]/(Sqrt[1+c*x]*Sqrt[-1+c*x])]*
    Int[(f*x)^(m+1)*(a+b*ArcCosh[c*x])^(n-1),x] -
  c^2/(f^2*(m+1))*Simp[Sqrt[d+e*x^2]/(Sqrt[1+c*x]*Sqrt[-1+c*x])]*
    Int[(f*x)^(m+2)*(a+b*ArcCosh[c*x])^n/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && LtQ[m,-1]
```

```
Int[(f_.*x_)^m_*Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]*(a+b*ArcCosh[c*x])^n/(f*(m+1)) -
  b*c*n/(f*(m+1))*Simp[Sqrt[d1+e1*x]/Sqrt[1+c*x]]*Simp[Sqrt[d2+e2*x]/Sqrt[-1+c*x]]*
    Int[(f*x)^(m+1)*(a+b*ArcCosh[c*x])^(n-1),x] -
  c^2/(f^2*(m+1))*Simp[Sqrt[d1+e1*x]/Sqrt[1+c*x]]*Simp[Sqrt[d2+e2*x]/Sqrt[-1+c*x]]*
    Int[((f*x)^(m+2)*(a+b*ArcCosh[c*x])^n)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && LtQ[m,-1]
```

2: $\int (f x)^m \sqrt{d+e x^2}\,(a+b\,\text{ArcCosh}[c x])^n\,dx$ when $c^2 d+e=0\land n\in\mathbb{Z}^+\land(m+2\in\mathbb{Z}^+\lor n=1)$

Derivation: Inverted integration by parts

Note: The piecewise constant factor in the second integral reduces the degree of d in the resulting antiderivative.

Rule: If $c^2 d+e=0\land n\in\mathbb{Z}^+\land(m+2\in\mathbb{Z}^+\lor n=1)$, then

$$\int (f x)^m \sqrt{d+e x^2}\,(a+b\,\text{ArcCosh}[c x])^n\,dx\longrightarrow$$

$$\frac{(f x)^{m+1}\sqrt{d+e x^2}\,(a+b\,\text{ArcCosh}[c x])^n}{f\,(m+2)}-$$

$$\frac{b c n\sqrt{d+e x^2}}{f\,(m+2)\sqrt{1+c x}\,\sqrt{-1+c x}}\int (f x)^{m+1}(a+b\,\text{ArcCosh}[c x])^{n-1}\,dx-\frac{\sqrt{d+e x^2}}{(m+2)\sqrt{1+c x}\,\sqrt{-1+c x}}\int \frac{(f x)^m (a+b\,\text{ArcCosh}[c x])^n}{\sqrt{1+c x}\,\sqrt{-1+c x}}\,dx$$

–

Program code:

```
Int[(f_.*x_)^m_*Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d+e*x^2]*(a+b*ArcCosh[c*x])^n/(f*(m+2)) -
  b*c*n/(f*(m+2))*Simp[Sqrt[d+e*x^2]/(Sqrt[1+c*x]*Sqrt[-1+c*x])]*
    Int[(f*x)^(m+1)*(a+b*ArcCosh[c*x])^(n-1),x] -
  1/(m+2)*Simp[Sqrt[d+e*x^2]/(Sqrt[1+c*x]*Sqrt[-1+c*x])]*
    Int[(f*x)^m*(a+b*ArcCosh[c*x])^n/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[c^2*d+e,0] && IGtQ[n,0] && (IGtQ[m,-2] || EqQ[n,1])
```

```
Int[(f_.*x_)^m_*Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]*(a+b*ArcCosh[c*x])^n/(f*(m+2)) -
  b*c*n/(f*(m+2))*Simp[Sqrt[d1+e1*x]/Sqrt[1+c*x]]*Simp[Sqrt[d2+e2*x]/Sqrt[-1+c*x]]*
    Int[(f*x)^(m+1)*(a+b*ArcCosh[c*x])^(n-1),x] -
  1/(m+2)*Simp[Sqrt[d1+e1*x]/Sqrt[1+c*x]]*Simp[Sqrt[d2+e2*x]/Sqrt[-1+c*x]]*
    Int[(f*x)^m*(a+b*ArcCosh[c*x])^n/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && IGtQ[n,0] && (IGtQ[m,-2] || EqQ[n,1])
```

3. $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \land n>0 \land p>0$

1: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \land n>0 \land p>0 \land m<-1$

Derivation: Inverted integration by parts

Rule: If $c^2 d+e=0 \land n>0 \land p>0 \land m<-1$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$\frac{(f x)^{m+1} \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n}{f (m+1)} - \frac{2 e p}{f^2 (m+1)} \int (f x)^{m+2} \left(d+e x^2\right)^{p-1} (a+b \operatorname{ArcCosh}[c x])^n \, dx - \frac{b c n \left(d+e x^2\right)^p}{f (m+1) (1+c x)^p (-1+c x)^p} \int (f x)^{m+1} (1+c x)^{p-\frac{1}{2}} (-1+c x)^{p-\frac{1}{2}} (a+b \operatorname{ArcCosh}[c x])^{n-1} \, dx$$

–

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n/(f*(m+1)) -
  2*e*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n/(f*(m+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m+1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && GtQ[p,0] && LtQ[m,-1]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n/(f*(m+1)) -
  2*e1*e2*p/(f^2*(m+1))*Int[(f*x)^(m+2)*(d1+e1*x)^(p-1)*(d2+e2*x)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n/(f*(m+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(f*x)^(m+1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && GtQ[p,0] && LtQ[m,-1]
```

x: $\int (f x)^m \left(d+e x^2\right)^p (a+b\,\text{ArcCosh}[c x])^n\,dx$ when $c^2 d+e=0 \land n>0 \land m>1 \land m+2p+1\neq 0 \land m\in\mathbb{Z}$

Rule: If $c^2 d+e=0 \land n>0 \land m>1 \land m+2p+1\neq 0 \land m\in\mathbb{Z}$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b\,\text{ArcCosh}[c x])^n\,dx \longrightarrow$$

$$\frac{f (f x)^{m-1} \left(d+e x^2\right)^{p+1} (a+b\,\text{ArcCosh}[c x])^n}{e\,(m+2p+1)}+\frac{f^2\,(m-1)}{c^2\,(m+2p+1)}\int (f x)^{m-2} \left(d+e x^2\right)^p (a+b\,\text{ArcCosh}[c x])^n\,dx-\frac{b f n \left(d+e x^2\right)^p}{c\,(m+2p+1)\,(1+c x)^p\,(-1+c x)^p}\int (f x)^{m-1} \left(-1+c^2 x^2\right)^{p+\frac{1}{2}} (a+b\,\text{ArcCosh}[c x])^{n-1}\,dx$$

Program code:

```
(* Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(e*(m+2*p+1)) +
  f^2*(m-1)/(c^2*(m+2*p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n/(c*(m+2*p+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m-1)*(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && EqQ[n,1] && IGtQ[p+1/2,0] && IGtQ[(m-1)/2,0] *)
```

x: $\int (f x)^m \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \wedge n>0 \wedge p<-1 \wedge m>1$

Derivation: Integration by parts

Basis: $x \left(d+e x^2\right)^p = \partial_x \frac{\left(d+e x^2\right)^{p+1}}{2 e (p+1)}$

Rule: If $c^2 d+e=0 \wedge n>0 \wedge p<-1 \wedge m>1$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^n \, dx \to$$

$$\frac{f (f x)^{m-1} \left(d+e x^2\right)^{p+1} (a+b \text{ArcCosh}[c x])^n}{2 e (p+1)} - \frac{f^2 (m-1)}{2 e (p+1)} \int (f x)^{m-2} \left(d+e x^2\right)^{p+1} (a+b \text{ArcCosh}[c x])^n \, dx - \frac{b f n \left(d+e x^2\right)^p}{2 c (p+1) (1+c x)^p (-1+c x)^p} \int (f x)^{m-1} (1+c x)^{p+\frac{1}{2}} (-1+c x)^{p+\frac{1}{2}} (a+b \text{ArcCosh}[c x])^{n-1} \, dx$$

–

Program code:

```
(* Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e*(p+1)) -
  f^2*(m-1)/(2*e*(p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n/(2*c*(p+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m-1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && EqQ[n,1] && ILtQ[p-1/2,0] && IGtQ[(m-1)/2,0] *)
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \wedge n>0 \wedge p>0 \wedge m \nless -1$

Derivation: Inverted integration by parts

Rule: If $c^2 d+e=0 \wedge n>0 \wedge p>0 \wedge m \nless -1$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$\frac{(f x)^{m+1} \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^n}{f (m+2 p+1)}+\frac{2 d p}{m+2 p+1} \int (f x)^m \left(d+e x^2\right)^{p-1} (a+b \text{ArcCosh}[c x])^n \, dx-\frac{b c n \left(d+e x^2\right)^p}{f (m+2 p+1) (1+c x)^p (-1+c x)^p} \int (f x)^{m+1} (1+c x)^{p-\frac{1}{2}} (-1+c x)^{p-\frac{1}{2}} (a+b \text{ArcCosh}[c x])^{n-1} \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n/(f*(m+2*p+1)) +
  2*d*p/(m+2*p+1)*Int[(f*x)^m*(d+e*x^2)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n/(f*(m+2*p+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m+1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && GtQ[p,0] && Not[LtQ[m,-1]]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n/(f*(m+2*p+1)) +
  2*d1*d2*p/(m+2*p+1)*Int[(f*x)^m*(d1+e1*x)^(p-1)*(d2+e2*x)^(p-1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n/(f*(m+2*p+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(f*x)^(m+1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && GtQ[p,0] && Not[LtQ[m,-1]]
```

4: $\int (f x)^m \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \land n>0 \land m+1\in\mathbb{Z}^-$

Rule: If $c^2 d+e=0 \land n>0 \land m+1\in\mathbb{Z}^-$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \text{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$\frac{(f x)^{m+1} \left(d+e x^2\right)^{p+1} (a+b \text{ArcCosh}[c x])^n}{d f (m+1)}+$$

$$\frac{c^2\,(m+2p+3)}{f^2\,(m+1)}\int (f\,x)^{m+2}\,(d+e\,x^2)^p\,(a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx\,+$$

$$\frac{b\,c\,n\,(d+e\,x^2)^p}{f\,(m+1)\,(1+c\,x)^p\,(-1+c\,x)^p}\int (f\,x)^{m+1}\,(1+c\,x)^{p+\frac{1}{2}}\,(-1+c\,x)^{p+\frac{1}{2}}\,(a+b\,\mathrm{ArcCosh}[c\,x])^{n-1}\,dx$$

–

Programcode:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(d*f*(m+1)) +
  c^2*(m+2*p+3)/(f^2*(m+1))*Int[(f*x)^(m+2)*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n,x] +
  b*c*n/(f*(m+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && ILtQ[m,-1]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  (f*x)^(m+1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(d1*d2*f*(m+1)) +
  c^2*(m+2*p+3)/(f^2*(m+1))*Int[(f*x)^(m+2)*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x] +
  b*c*n/(f*(m+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,p},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && ILtQ[m,-1]
```

5. $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx$ when $c^2\,d+e==0\,\land\,n>0\,\land\,p<-1\,\land\,m\in\mathbb{Z}$

1: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx$ when $c^2\,d+e==0\,\land\,n>0\,\land\,p<-1\,\land\,m-1\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: $x\,(d+e\,x^2)^p == \partial_x \frac{(d+e\,x^2)^{p+1}}{2\,e\,(p+1)}$

Rule: If $c^2\,d+e==0\,\land\,n>0\,\land\,p<-1\,\land\,m-1\in\mathbb{Z}^+$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\mathrm{ArcCosh}[c\,x])^n\,dx\,\longrightarrow$$

$$\frac{f\,(f\,x)^{m-1}\,(d+e\,x^2)^{p+1}\,(a+b\,\mathrm{ArcCosh}[c\,x])^n}{2\,e\,(p+1)}\,-$$

$$\frac{f^2\,(m-1)}{2\,e\,(p+1)}\int (f\,x)^{m-2}\left(d+e\,x^2\right)^{p+1}(a+b\,\text{ArcCosh}[c\,x])^n\,dx-$$

$$\frac{b\,f\,n\left(d+e\,x^2\right)^p}{2\,c\,(p+1)\,(1+c\,x)^p\,(-1+c\,x)^p}\int (f\,x)^{m-1}\,(1+c\,x)^{p+\frac{1}{2}}\,(-1+c\,x)^{p+\frac{1}{2}}\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}\,dx$$

–

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e*(p+1)) -
  f^2*(m-1)/(2*e*(p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n/(2*c*(p+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m-1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && LtQ[p,-1] && IGtQ[m,1]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*e1*e2*(p+1)) -
  f^2*(m-1)/(2*e1*e2*(p+1))*Int[(f*x)^(m-2)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n/(2*c*(p+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(f*x)^(m-1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && LtQ[p,-1] && IGtQ[m,1]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \land n>0 \land p<-1 \land m \in \mathbb{Z}^-$

Rule: If $c^2 d+e=0 \land n>0 \land p<-1 \land m \in \mathbb{Z}^-$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$-\frac{(f x)^{m+1} \left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcCosh}[c x])^n}{2 d f (p+1)}+$$
$$\frac{m+2 p+3}{2 d (p+1)} \int (f x)^m \left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcCosh}[c x])^n \, dx-$$
$$\frac{b c n \left(d+e x^2\right)^p}{2 f (p+1) (1+c x)^p (-1+c x)^p} \int (f x)^{m+1} (1+c x)^{p+\frac{1}{2}} (-1+c x)^{p+\frac{1}{2}} (a+b \operatorname{ArcCosh}[c x])^{n-1} \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  -(f*x)^(m+1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*d*f*(p+1)) +
  (m+2*p+3)/(2*d*(p+1))*Int[(f*x)^m*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n/(2*f*(p+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && LtQ[p,-1] && Not[GtQ[m,1]] && (IntegerQ[m] || IntegerQ[p] || EqQ[n,1])
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  -(f*x)^(m+1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(2*d1*d2*f*(p+1)) +
  (m+2*p+3)/(2*d1*d2*(p+1))*Int[(f*x)^m*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n,x] -
  b*c*n/(2*f*(p+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(f*x)^(m+1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && LtQ[p,-1] && Not[GtQ[m,1]] && (IntegerQ[m] || EqQ[n,1])
```

6: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \land n>0 \land m-1\in\mathbb{Z}^+ \land m+2p+1\neq 0$

Derivation: Inverted integration by parts

Rule: If $c^2 d+e=0 \land n>0 \land m-1\in\mathbb{Z}^+ \land m+2p+1\neq 0$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$\frac{f (f x)^{m-1} \left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcCosh}[c x])^n}{e (m+2p+1)}+\frac{f^2 (m-1)}{c^2 (m+2p+1)} \int (f x)^{m-2} \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx-\frac{b f n \left(d+e x^2\right)^p}{c (m+2p+1) (1+c x)^p (-1+c x)^p} \int (f x)^{m-1} (1+c x)^{p+\frac{1}{2}} (-1+c x)^{p+\frac{1}{2}} (a+b \operatorname{ArcCosh}[c x])^{n-1} \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])^n/(e*(m+2*p+1)) +
  f^2*(m-1)/(c^2*(m+2*p+1))*Int[(f*x)^(m-2)*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n/(c*(m+2*p+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m-1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d,e,f,p},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && IGtQ[m,1] && NeQ[m+2*p+1,0]
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  f*(f*x)^(m-1)*(d1+e1*x)^(p+1)*(d2+e2*x)^(p+1)*(a+b*ArcCosh[c*x])^n/(e1*e2*(m+2*p+1)) +
  f^2*(m-1)/(c^2*(m+2*p+1))*Int[(f*x)^(m-2)*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x] -
  b*f*n/(c*(m+2*p+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(f*x)^(m-1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n-1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,p},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && IGtQ[m,1] && NeQ[m+2*p+1,0]
```

2. $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $c^2\,d+e=0\,\wedge\,n<-1$

1: $\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $c^2\,d+e=0\,\wedge\,n<-1\,\wedge\,m+2\,p+1=0$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}} = \partial_x \frac{(a+b\,\text{ArcCosh}[c\,x])^{n+1}}{b\,c\,(n+1)}$

Basis: If $c^2\,d+e=0\,\wedge\,m+2\,p+1=0$, then $\partial_x\left((f\,x)^m\sqrt{1+c\,x}\,\sqrt{-1+c\,x}\,\left(d+e\,x^2\right)^p\right) = -\frac{f\,m\,(f\,x)^{m-1}\left(d+e\,x^2\right)^p}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}$

Basis: If $c^2\,d+e=0$, then $\partial_x \frac{\left(d+e\,x^2\right)^p}{(1+c\,x)^p\,(-1+c\,x)^p} = 0$

Rule: If $c^2\,d+e=0\,\wedge\,n<-1\,\wedge\,m+2\,p+1=0$, then

$$\int (f\,x)^m \left(d+e\,x^2\right)^p (a+b\,\text{ArcCosh}[c\,x])^n\,dx \longrightarrow$$

$$\frac{(f\,x)^m\sqrt{1+c\,x}\,\sqrt{-1+c\,x}\,\left(d+e\,x^2\right)^p (a+b\,\text{ArcCosh}[c\,x])^{n+1}}{b\,c\,(n+1)} +$$

$$\frac{f\,m\left(d+e\,x^2\right)^p}{b\,c\,(n+1)\,(1+c\,x)^p\,(-1+c\,x)^p}\int (f\,x)^{m-1}\,(1+c\,x)^{p-\frac{1}{2}}\,(-1+c\,x)^{p-\frac{1}{2}}\,(a+b\,\text{ArcCosh}[c\,x])^{n+1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]*(d+e*x^2)^p]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  f*m/(b*c*(n+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m-1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[c^2*d+e,0] && LtQ[n,-1] && EqQ[m+2*p+1,0]
```

```
Int[(f_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]*(d1+e1*x)^p]*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  f*m/(b*c*(n+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(f*x)^(m-1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,p},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && LtQ[n,-1] && EqQ[m+2*p+1,0]
```

2: $\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $c^2 d+e=0 \land n<-1 \land 2p \in \mathbb{Z}^+ \land m+2p+1 \neq 0$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{(a+b \operatorname{ArcCosh}[c x])^n}{\sqrt{1+c x}\,\sqrt{-1+c x}} = \partial_x \frac{(a+b \operatorname{ArcCosh}[c x])^{n+1}}{b c (n+1)}$

Basis: If $c^2 d+e=0$, then

$$\partial_x \left((f x)^m \sqrt{1+c x}\,\sqrt{-1+c x}\,\left(d+e x^2\right)^p\right) = -\frac{f m (f x)^{m-1}\left(d+e x^2\right)^p}{\sqrt{1+c x}\,\sqrt{-1+c x}} + \frac{c^2 (m+2p+1)(f x)^{m+1}\left(d+e x^2\right)^p}{f\sqrt{1+c x}\,\sqrt{-1+c x}}$$

Basis: If $c^2 d+e=0$, then $\partial_x \frac{\left(d+e x^2\right)^p}{(1+c x)^p(-1+c x)^p} = 0$

Basis: If $p+\frac{1}{2} \in \mathbb{Z}$, then $(1+c x)^{p-\frac{1}{2}}(-1+c x)^{p-\frac{1}{2}} = \left(-1+c^2 x^2\right)^{p-\frac{1}{2}}$

– Rule: If $c^2 d+e=0 \land n<-1 \land 2p \in \mathbb{Z}^+ \land m+2p+1 \neq 0$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow$$

$$\frac{(f x)^m \sqrt{1+c x}\,\sqrt{-1+c x}\,\left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^{n+1}}{b c (n+1)} +$$

$$\frac{f m}{b c (n+1)} \int \frac{(f x)^{m-1}\left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^{n+1}}{\sqrt{1+c x}\,\sqrt{-1+c x}} \, dx -$$

$$\frac{c (m+2p+1)}{b f (n+1)} \int \frac{(f x)^{m+1}\left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^{n+1}}{\sqrt{1+c x}\,\sqrt{-1+c x}} \, dx \longrightarrow$$

$$\frac{(f x)^m \sqrt{1+c x}\,\sqrt{-1+c x}\,\left(d+e x^2\right)^p (a+b\,\text{ArcCosh}[c x])^{n+1}}{b c\,(n+1)}+$$
$$\frac{f m\left(d+e x^2\right)^p}{b c\,(n+1)\,(1+c x)^p\,(-1+c x)^p}\int (f x)^{m-1} (1+c x)^{p-\frac{1}{2}} (-1+c x)^{p-\frac{1}{2}} (a+b\,\text{ArcCosh}[c x])^{n+1}\,dx-$$
$$\frac{c\,(m+2 p+1)\left(d+e x^2\right)^p}{b f\,(n+1)\,(1+c x)^p\,(-1+c x)^p}\int (f x)^{m+1} (1+c x)^{p-\frac{1}{2}} (-1+c x)^{p-\frac{1}{2}} (a+b\,\text{ArcCosh}[c x])^{n+1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]*(d+e*x^2)^p]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  f*m/(b*c*(n+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m-1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] -
  c*(m+2*p+1)/(b*f*(n+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m+1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[c^2*d+e,0] && LtQ[n,-1] && IGtQ[2*p,0] && NeQ[m+2*p+1,0] && IGtQ[m,-3]
```

```
Int[(f_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Sqrt[1+c*x]*Sqrt[-1+c*x]*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) +
  f*m/(b*c*(n+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
     Int[(f*x)^(m-1)*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] -
  c*(m+2*p+1)/(b*f*(n+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(f*x)^(m+1)*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,p},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && LtQ[n,-1] && IGtQ[p+1/2,0] && NeQ[m+2*p+1,0] && IGtQ[m,-3]
```

3: $\int (f x)^m \left(d+e x^2\right)^p (a+b\,\text{ArcCosh}[c x])^n\,dx$ when $c^2 d+e=0 \land n<-1 \land p \not> 0 \land p \neq -\frac{1}{2}$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{(a+b\,\text{ArcCosh}[c x])^n}{\sqrt{1+c x}\,\sqrt{-1+c x}} = \partial_x \frac{(a+b\,\text{ArcCosh}[c x])^{n+1}}{b c\,(n+1)}$

Basis: If $c^2 d + e == 0$, then $\partial_x \left((f x)^m \sqrt{1+c x} \sqrt{-1+c x} \left(d+e x^2\right)^p\right) ==$

$$f m (f x)^{m-1} \sqrt{1+c x} \sqrt{-1+c x} \left(d+e x^2\right)^p + \frac{c^2 (2 p+1) (f x)^{m+1} \left(d+e x^2\right)^p}{f \sqrt{1+c x} \sqrt{-1+c x}}$$

Basis: If $c^2 d + e == 0$, then $\partial_x \frac{\left(d+e x^2\right)^p}{(1+c x)^p (-1+c x)^p} == 0$

Basis: If $p + \frac{1}{2} \in \mathbb{Z}$, then $(1+c x)^{p-\frac{1}{2}} (-1+c x)^{p-\frac{1}{2}} == \left(-1+c^2 x^2\right)^{p-\frac{1}{2}}$

–

Rule: If $c^2 d + e == 0 \land n < -1 \land p \not> 0 \land p \neq -\frac{1}{2}$, then

$$\int (f x)^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$$

$$\rightarrow \frac{(f x)^m \sqrt{1+c x} \sqrt{-1+c x} \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^{n+1}}{b c (n+1)} -$$

$$\frac{f m}{b c (n+1)} \int (f x)^{m-1} \sqrt{1+c x} \sqrt{-1+c x} \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^{n+1} \, dx -$$

$$\frac{c (2 p+1)}{b f (n+1)} \int \frac{(f x)^{m+1} \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^{n+1}}{\sqrt{1+c x} \sqrt{-1+c x}} \, dx$$

$$\rightarrow \frac{(f x)^m \sqrt{1+c x} \sqrt{-1+c x} \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^{n+1}}{b c (n+1)} -$$

$$\frac{f m \left(d+e x^2\right)^p}{b c (n+1) (1+c x)^p (-1+c x)^p} \int (f x)^{m-1} (1+c x)^{p+\frac{1}{2}} (-1+c x)^{p+\frac{1}{2}} (a+b \operatorname{ArcCosh}[c x])^{n+1} \, dx -$$

$$\frac{c (2 p+1) \left(d+e x^2\right)^p}{b f (n+1) (1+c x)^p (-1+c x)^p} \int (f x)^{m+1} (1+c x)^{p-\frac{1}{2}} (-1+c x)^{p-\frac{1}{2}} (a+b \operatorname{ArcCosh}[c x])^{n+1} \, dx$$

Program code:

```
(* Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]*(d+e*x^2)^p]*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) -
  f*m/(b*c*(n+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m-1)*(1+c*x)^(p+1/2)*(-1+c*x)^(p+1/2)*(a+b*ArcCosh[c*x])^(n+1),x] -
  c*(2*p+1)/(b*f*(n+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Int[(f*x)^(m+1)*(1+c*x)^(p-1/2)*(-1+c*x)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && EqQ[c^2*d+e,0] && LtQ[n,-1] && NeQ[p,-1/2] && IntegerQ[2*p] && IGtQ[m,-3] *)
```

```
(* Int[(f_.*x_)^m_.*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  (f*x)^m*Sqrt[1+c*x]*Sqrt[-1+c*x]*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1)) -
  f*m/(b*c*(n+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(f*x)^(m-1)*(-1+c^2*x^2)^(p+1/2)*(a+b*ArcCosh[c*x])^(n+1),x] -
  c*(2*p+1)/(b*f*(n+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Int[(f*x)^(m+1)*(-1+c^2*x^2)^(p-1/2)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,p},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && LtQ[n,-1] && ILtQ[p+1/2,0] && IGtQ[m,-3] *)
```

3. $\int \frac{(f\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $c^2\,d+e==0$

1. $\int \frac{(f\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $c^2\,d+e==0\,\land\,n>0$

1: $\int \frac{(f\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $c^2\,d+e==0\,\land\,n>0\,\land\,m-1\in\mathbb{Z}^+$

Rule: If $c^2\,d+e==0\,\land\,n>0\,\land\,m-1\in\mathbb{Z}^+$, then

$$\int \frac{(f\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx \longrightarrow$$

$$\frac{f\,(f\,x)^{m-1}\,\sqrt{d+e\,x^2}\,(a+b\,\text{ArcCosh}[c\,x])^n}{e\,m} -$$

$$\frac{b\,f\,n\,\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{c\,m\,\sqrt{d+e\,x^2}}\int (f\,x)^{m-1}\,(a+b\,\text{ArcCosh}[c\,x])^{n-1}\,dx+\frac{f^2\,(m-1)}{c^2\,m}\int \frac{(f\,x)^{m-2}\,(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$$

–

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  f*(f*x)^(m-1)*Sqrt[d+e*x^2]*(a+b*ArcCosh[c*x])^n/(e*m) -
  b*f*n/(c*m)*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]/Sqrt[d+e*x^2]]*Int[(f*x)^(m-1)*(a+b*ArcCosh[c*x])^(n-1),x] +
  f^2*(m-1)/(c^2*m)*Int[(f*x)^(m-2)*(a+b*ArcCosh[c*x])^n/Sqrt[d+e*x^2],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[c^2*d+e,0] && GtQ[n,0] && IGtQ[m,1]
```

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  f*(f*x)^(m-1)*Sqrt[d1+e1*x]*Sqrt[d2+e2*x]*(a+b*ArcCosh[c*x])^n/(e1*e2*m) -
  b*f*n/(c*m)*Simp[Sqrt[1+c*x]/Sqrt[d1+e1*x]]*Simp[Sqrt[-1+c*x]/Sqrt[d2+e2*x]]*
    Int[(f*x)^(m-1)*(a+b*ArcCosh[c*x])^(n-1),x] +
  f^2*(m-1)/(c^2*m)*Int[(f*x)^(m-2)*(a+b*ArcCosh[c*x])^n/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x]),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[n,0] && IGtQ[m,1]
```

2: $\int \frac{x^m (a + b \text{ArcCosh}[c x])^n}{\sqrt{d + e x^2}} dx$ when $c^2 d + e == 0 \land n \in \mathbb{Z}^+ \land m \in \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

– Basis: If $c^2 d + e == 0$, then $\partial_x \frac{\sqrt{1+c x} \sqrt{-1+c x}}{\sqrt{d+e x^2}} == 0$

Basis: If $m \in \mathbb{Z}$, then $\frac{x^m}{\sqrt{1+c x} \sqrt{-1+c x}} == \frac{1}{c^{m+1}} \text{Subst}[\text{Cosh}[x]^m, x, \text{ArcCosh}[c x]] \partial_x \text{ArcCosh}[c x]$

– Note: If $n \in \mathbb{Z}^+$, then $(a + b x)^n \text{Cosh}[x]$ is integrable in closed-form.

Rule: If $c^2 d + e == 0 \land n \in \mathbb{Z}^+ \land m \in \mathbb{Z}$, then

$$\int \frac{x^m (a + b \text{ArcCosh}[c x])^n}{\sqrt{d + e x^2}} dx \to \frac{\sqrt{1 + c x} \sqrt{-1 + c x}}{c^{m+1} \sqrt{d + e x^2}} \text{Subst}\left[\int (a + b x)^n \text{Cosh}[x]^m dx, x, \text{ArcCosh}[c x]\right]$$

Program code:

```
Int[x_^m_*(a_.+b_.*ArcCosh[c_.*x_])^n_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/c^(m+1)*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]/Sqrt[d+e*x^2]]*
    Subst[Int[(a+b*x)^n*Cosh[x]^m,x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c^2*d+e,0] && IGtQ[n,0] && IntegerQ[m]
```

```
Int[x_^m_*(a_.+b_.*ArcCosh[c_.*x_])^n_./(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  1/c^(m+1)*Simp[Sqrt[1+c*x]/Sqrt[d1+e1*x]]*Simp[Sqrt[-1+c*x]/Sqrt[d2+e2*x]]*
    Subst[Int[(a+b*x)^n*Cosh[x]^m,x],x,ArcCosh[c*x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && IGtQ[n,0] && IntegerQ[m]
```

3: $\int \frac{(f x)^m (a + b \text{ArcCosh}[c x])}{\sqrt{d + e x^2}} dx$ when $c^2 d + e == 0 \land m \notin \mathbb{Z}$

– Rule: If $c^2 d + e == 0 \land m \notin \mathbb{Z}$, then

$$\int \frac{(f x)^m (a + b \text{ArcCosh}[c x])}{\sqrt{d + e x^2}} dx \to$$

$$\frac{(f\,x)^{m+1}\,\sqrt{1-c^2\,x^2}\,\left(a+b\,\text{ArcCosh}[c\,x]\right)}{f\,(m+1)\,\sqrt{d+e\,x^2}}\,\text{Hypergeometric2F1}\left[\frac{1}{2},\frac{1+m}{2},\frac{3+m}{2},c^2\,x^2\right]+$$
$$\frac{b\,c\,(f\,x)^{m+2}\,\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{f^2\,(m+1)\,(m+2)\,\sqrt{d+e\,x^2}}\,\text{HypergeometricPFQ}\left[\left\{1,1+\frac{m}{2},1+\frac{m}{2}\right\},\left\{\frac{3}{2}+\frac{m}{2},2+\frac{m}{2}\right\},c^2\,x^2\right]$$

–

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_])/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  (f*x)^(m+1)/(f*(m+1))*Simp[Sqrt[1-c^2*x^2]/Sqrt[d+e*x^2]]*
    (a+b*ArcCosh[c*x])*Hypergeometric2F1[1/2,(1+m)/2,(3+m)/2,c^2*x^2] +
  b*c*(f*x)^(m+2)/(f^2*(m+1)*(m+2))*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]/Sqrt[d+e*x^2]]*
    HypergeometricPFQ[{1,1+m/2,1+m/2},{3/2+m/2,2+m/2},c^2*x^2] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[c^2*d+e,0] && Not[IntegerQ[m]]
```

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCosh[c_.*x_])/(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  (f*x)^(m+1)/(f*(m+1))*Simp[Sqrt[1-c^2*x^2]/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x])]*
    (a+b*ArcCosh[c*x])*Hypergeometric2F1[1/2,(1+m)/2,(3+m)/2,c^2*x^2] +
  b*c*(f*x)^(m+2)/(f^2*(m+1)*(m+2))*Simp[Sqrt[1+c*x]/Sqrt[d1+e1*x]]*Simp[Sqrt[-1+c*x]/Sqrt[d2+e2*x]]*
    HypergeometricPFQ[{1,1+m/2,1+m/2},{3/2+m/2,2+m/2},c^2*x^2] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && Not[IntegerQ[m]]
```

2: $\int \frac{(f\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$ when $c^2\,d+e=0\,\wedge\,n<-1$

Derivation: Integration by parts and piecewise constant extraction

Basis: $\frac{(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}=\partial_x\frac{(a+b\,\text{ArcCosh}[c\,x])^{n+1}}{b\,c\,(n+1)}$

Basis: If $c^2\,d+e=0$, then $\partial_x\frac{(f\,x)^m\,\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{\sqrt{d+e\,x^2}}=\frac{f\,m\,(f\,x)^{m-1}\,\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{\sqrt{d+e\,x^2}}$

Basis: If $c^2\,d+e=0$, then $\partial_x\frac{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{\sqrt{d+e\,x^2}}=0$

Rule: If $c^2\,d+e=0\,\wedge\,n<-1$, then

$$\int\frac{(f\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^n}{\sqrt{d+e\,x^2}}\,dx$$

$$\rightarrow\frac{(f\,x)^m\,(a+b\,\text{ArcCosh}[c\,x])^{n+1}\,\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{b\,c\,(n+1)\,\sqrt{d+e\,x^2}}-\frac{f\,m\,\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}{b\,c\,(n+1)\,\sqrt{d+e\,x^2}}\int(f\,x)^{m-1}\,(a+b\,\text{ArcCosh}[c\,x])^{n+1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  (f*x)^m*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1))*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]/Sqrt[d+e*x^2]] -
  f*m/(b*c*(n+1))*Simp[Sqrt[1+c*x]*Sqrt[-1+c*x]/Sqrt[d+e*x^2]]*Int[(f*x)^(m-1)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[c^2*d+e,0] && LtQ[n,-1]
```

```
Int[(f_.*x_)^m_.*(a_.+b_.*ArcCosh[c_.*x_])^n_/(Sqrt[d1_+e1_.*x_]*Sqrt[d2_+e2_.*x_]),x_Symbol] :=
  (f*x)^m*(a+b*ArcCosh[c*x])^(n+1)/(b*c*(n+1))*Simp[Sqrt[1+c*x]/Sqrt[d1+e1*x]]*Simp[Sqrt[-1+c*x]/Sqrt[d2+e2*x]] -
  f*m/(b*c*(n+1))*Simp[Sqrt[1+c*x]/Sqrt[d1+e1*x]]*Simp[Sqrt[-1+c*x]/Sqrt[d2+e2*x]]*
    Int[(f*x)^(m-1)*(a+b*ArcCosh[c*x])^(n+1),x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && LtQ[n,-1]
```

4: $\int x^m \left(d+e x^2\right)^p (a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $c^2 d+e==0 \land 2p+2\in\mathbb{Z}^+ \land m\in\mathbb{Z}^+$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $c^2 d+e==0$, then $\partial_x \frac{\left(d+e x^2\right)^p}{(1+c x)^p (-1+c x)^p}==0$

Basis: If $2p\in\mathbb{Z} \land m\in\mathbb{Z}$, then
$x^m (1+c x)^p (-1+c x)^p ==$
$\frac{1}{b\,c^{m+1}}\,\text{Subst}\left[\text{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^m \text{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right]^{2p+1},\,x,\,a+b\,\text{ArcCosh}[c\,x]\right]\,\partial_x\,(a+b\,\text{ArcCosh}[c\,x])$

Note: If $2p+2\in\mathbb{Z}^+ \land m\in\mathbb{Z}^+$, then $x^n\,\text{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^m \text{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right]^{2p+1}$ is integrable in closed-form.

Rule: If $c^2 d+e==0 \land 2p+2\in\mathbb{Z}^+ \land m\in\mathbb{Z}^+$, then

$$\int x^m \left(d+e x^2\right)^p (a+b\,\text{ArcCosh}[c\,x])^n\,dx$$

$$\rightarrow \frac{\left(d+e x^2\right)^p}{(1+c x)^p (-1+c x)^p}\int x^m (1+c x)^p (-1+c x)^p (a+b\,\text{ArcCosh}[c\,x])^n\,dx$$

$$\rightarrow \frac{\left(d+e x^2\right)^p}{b\,c^{m+1} (1+c x)^p (-1+c x)^p}\,\text{Subst}\left[\int x^n\,\text{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^m \text{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right]^{2p+1} dx,\,x,\,a+b\,\text{ArcCosh}[c\,x]\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  1/(b*c^(m+1))*Simp[(d+e*x^2)^p/((1+c*x)^p*(-1+c*x)^p)]*
    Subst[Int[x^n*Cosh[-a/b+x/b]^m*Sinh[-a/b+x/b]^(2*p+1),x],x,a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[c^2*d+e,0] && IGtQ[2*p+2,0] && IGtQ[m,0]
```

```
Int[x_^m_.*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  1/(b*c^(m+1))*Simp[(d1+e1*x)^p/(1+c*x)^p]*Simp[(d2+e2*x)^p/(-1+c*x)^p]*
    Subst[Int[x^n*Cosh[-a/b+x/b]^m*Sinh[-a/b+x/b]^(2*p+1),x],x,a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c,d1,e1,d2,e2,n},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && IGtQ[p+3/2,0] && IGtQ[m,0]
```

5: $\int (f x)^m (d + e x^2)^p (a + b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d + e == 0 \land p + \frac{1}{2} \in \mathbb{Z}^+ \land \frac{m+1}{2} \notin \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $c^2 d + e == 0 \land p + \frac{1}{2} \in \mathbb{Z}^+ \land \frac{m+1}{2} \notin \mathbb{Z}^+$, then

$$\int (f x)^m (d + e x^2)^p (a + b \text{ArcCosh}[c x])^n \, dx \to \int \frac{(a + b \text{ArcCosh}[c x])^n}{\sqrt{d + e x^2}} \text{ExpandIntegrand}\left[(f x)^m (d + e x^2)^{p+\frac{1}{2}}, x\right] dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCosh[c*x])^n/Sqrt[d+e*x^2],(f*x)^m*(d+e*x^2)^(p+1/2),x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[c^2*d+e,0] && IGtQ[p+1/2,0] && Not[IGtQ[(m+1)/2,0]] && (EqQ[m,-1] || EqQ[m,-2])
```

```
Int[(f_.*x_)^m_*(d1_+e1_.*x_)^p_*(d2_+e2_.*x_)^p_*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCosh[c*x])^n/(Sqrt[d1+e1*x]*Sqrt[d2+e2*x]),(f*x)^m*(d1+e1*x)^(p+1/2)*(d2+e2*x)^(p+1/2),x],x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,n},x] && EqQ[e1,c*d1] && EqQ[e2,-c*d2] && GtQ[d1,0] && LtQ[d2,0] && IGtQ[p+1/2,0] && Not[IGtQ[(m+1)/2,0]] &&
  (EqQ[m,-1] || EqQ[m,-2])
```

2. $\int (f x)^m (d + e x^2)^p (a + b \text{ArcCosh}[c x])^n \, dx$ when $c^2 d + e \neq 0$

0: $\int (f x)^m (d + e x^2) (a + b \text{ArcCosh}[c x]) \, dx$ when $c^2 d + e \neq 0 \land m \neq -1 \land m \neq -3$

Derivation: Integration by parts

Note: This rule can be removed when integrands of the form $(d + e x)^m (f + g x)^m \left(a + c x^2\right)^p$ when $e f + d g == 0$ are integrated without first resorting to piecewise constant extraction.

Rule: If $c^2 d + e \neq 0 \land m \neq -1 \land m \neq -3$, then

$$\int (f x)^m \left(d+e x^2\right) (a+b \operatorname{ArcCosh}[c x])\, dx \longrightarrow$$
$$\frac{d (f x)^{m+1} (a+b \operatorname{ArcCosh}[c x])}{f (m+1)}+\frac{e (f x)^{m+3} (a+b \operatorname{ArcCosh}[c x])}{f^3 (m+3)}-\frac{b c}{f (m+1) (m+3)} \int \frac{(f x)^{m+1} \left(d (m+3)+e (m+1) x^2\right)}{\sqrt{1+c x} \sqrt{-1+c x}}\, dx$$

– Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  d*(f*x)^(m+1)*(a+b*ArcCosh[c*x])/(f*(m+1)) +
  e*(f*x)^(m+3)*(a+b*ArcCosh[c*x])/(f^3*(m+3)) -
  b*c/(f*(m+1)*(m+3))*Int[(f*x)^(m+1)*(d*(m+3)+e*(m+1)*x^2)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[c^2*d+e,0] && NeQ[m,-1] && NeQ[m,-3]
```

1: $\int x \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])\, dx$ when $c^2 d+e \neq 0 \land p \neq -1$

– Derivation: Integration by parts

– Basis:: If $p \neq -1$, then $x \left(d+e x^2\right)^p = \partial_x \frac{\left(d+e x^2\right)^{p+1}}{2 e (p+1)}$

– Rule: If $c^2 d+e \neq 0 \land p \neq -1$, then

$$\int x \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])\, dx \longrightarrow \frac{\left(d+e x^2\right)^{p+1} (a+b \operatorname{ArcCosh}[c x])}{2 e (p+1)}-\frac{b c}{2 e (p+1)} \int \frac{\left(d+e x^2\right)^{p+1}}{\sqrt{1+c x} \sqrt{-1+c x}}\, dx$$

– Program code:

```
Int[x_*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  (d+e*x^2)^(p+1)*(a+b*ArcCosh[c*x])/(2*e*(p+1)) - b*c/(2*e*(p+1))*Int[(d+e*x^2)^(p+1)/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[c^2*d+e,0] && NeQ[p,-1]
```

2: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcCosh}[c\,x])\,dx$ when $c^2\,d+e\neq 0\,\wedge\,p\in\mathbb{Z}\,\wedge\,\left(p>0\,\vee\,\frac{m-1}{2}\in\mathbb{Z}^+\wedge\,m+p\leq 0\right)$

Derivation: Integration by parts

Note: If $\frac{m-1}{2}\in\mathbb{Z}^+\wedge\,p\in\mathbb{Z}^-\wedge\,m+p\geq 0$, then $\int (f\,x)^m\,(d+e\,x^2)^p$ is a rational function.

Rule: If $c^2\,d+e\neq 0\,\wedge\,p\in\mathbb{Z}\,\wedge\,\left(p>0\,\vee\,\frac{m-1}{2}\in\mathbb{Z}^+\wedge\,m+p\leq 0\right)$, let $u\to\int (f\,x)^m\,(d+e\,x^2)^p\,dx$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcCosh}[c\,x])\,dx\;\longrightarrow\;u\,(a+b\,\text{ArcCosh}[c\,x])-b\,c\int\frac{u}{\sqrt{1+c\,x}\,\sqrt{-1+c\,x}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^p,x]},
  Dist[a+b*ArcCosh[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/(Sqrt[1+c*x]*Sqrt[-1+c*x]),x],x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && NeQ[c^2*d+e,0] && IntegerQ[p] && (GtQ[p,0] || IGtQ[(m-1)/2,0] && LeQ[m+p,0])
```

x: $\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx$ when $m \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x] == \frac{1}{b c} \operatorname{Subst}\left[F\left[\frac{\operatorname{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]}{c}\right] \operatorname{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right], x, a+b \operatorname{ArcCosh}[c x]\right] \partial_x (a+b \operatorname{ArcCosh}[c x])$

Note: If $m \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+$, then $x^n \operatorname{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^m \left(c^2 d+e \operatorname{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^2\right)^p \operatorname{Sinh}\left[\frac{a}{b}-\frac{x}{b}\right]$ is integrable in closed-form.

Rule: If $m \in \mathbb{Z}^+ \land p \in \mathbb{Z}^+$, then

$$\int x^m \left(d+e x^2\right)^p (a+b \operatorname{ArcCosh}[c x])^n \, dx \longrightarrow \frac{1}{b c^{m+2 p+1}} \operatorname{Subst}\left[\int x^n \operatorname{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^m \left(c^2 d+e \operatorname{Cosh}\left[-\frac{a}{b}+\frac{x}{b}\right]^2\right)^p \operatorname{Sinh}\left[-\frac{a}{b}+\frac{x}{b}\right] dx, x, a+b \operatorname{ArcCosh}[c x]\right]$$

Program code:

```
(* Int[x_^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_,x_Symbol] :=
  1/(b*c^(m+2*p+1))*Subst[Int[x^n*Cosh[-a/b+x/b]^m*(c^2*d+e*Cosh[-a/b+x/b]^2)^p*Sinh[-a/b+x/b],x],x,a+b*ArcCosh[c*x]] /;
FreeQ[{a,b,c,d,e,n},x] && IGtQ[m,0] && IGtQ[p,0] *)
```

3: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$ when $c^2\,d+e\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p\in\mathbb{Z}\,\wedge\,m\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $c^2\,d+e\neq 0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p\in\mathbb{Z}\,\wedge\,m\in\mathbb{Z}$, then

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx\,\longrightarrow\,\int (a+b\,\text{ArcCosh}[c\,x])^n\,\text{ExpandIntegrand}\left[(f\,x)^m\,(d+e\,x^2)^p,\,x\right]\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCosh[c*x])^n,(f*x)^m*(d+e*x^2)^p,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && NeQ[c^2*d+e,0] && IGtQ[n,0] && IntegerQ[p] && IntegerQ[m]
```

U: $\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$

Rule:

$$\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx\,\longrightarrow\,\int (f\,x)^m\,(d+e\,x^2)^p\,(a+b\,\text{ArcCosh}[c\,x])^n\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d+e*x^2)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

```
Int[(f_.*x_)^m_.*(d1_+e1_.*x_)^p_.*(d2_+e2_.*x_)^p_.*(a_.+b_.*ArcCosh[c_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(f*x)^m*(d1+e1*x)^p*(d2+e2*x)^p*(a+b*ArcCosh[c*x])^n,x] /;
FreeQ[{a,b,c,d1,e1,d2,e2,f,m,n,p},x]
```